
Tom Hales is famous for many things . . .



. . . including the first Jordan Curve Theorem formalization

https://en.wikipedia.org/wiki/Jordan_curve_theorem

https://en.wikipedia.org/wiki/Jordan_curve_theorem


. . . and the rehabilitation of the original proof

https://en.wikipedia.org/wiki/Jordan_curve_theorem

https://en.wikipedia.org/wiki/Jordan_curve_theorem


What is the Jordan Curve Theorem?

Every simple closed curve in the plane separates the plane into
exactly two connected subsets, one bounded (the “inside”) and
one unbounded (the “outside”).

I Curve: continuous function γ : [0, 1]→ R2

I Closed: γ(0) = γ(1)

I Simple: ∀x y . γ(x) = γ(y)⇒ x = y ∨ {x , y} = {0, 1}



Alternative point of view

Alternatively: a simple closed curve is a homeomorphic map
γ : S1 → R2 out of the unit circle S1 = {x ∈ R2 | |x | = 1}

Every subset of R2 homeomorphic to S1 separates the plane into
exactly two connected subsets, one bounded (the “inside”) and
one unbounded (the “outside”).



The Jordan Curve Theorem in HOL Light

|- !c. simple_path c /\ pathfinish c = pathstart c

==> ?ins out.

~(ins = {}) /\ open ins /\ connected ins /\

~(out = {}) /\ open out /\ connected out /\

bounded ins /\ ~bounded out /\

ins INTER out = {} /\

ins UNION out =

(:real^2) DIFF path_image c /\

frontier ins = path_image c /\

frontier out = path_image c



The Jordan-Schoenflies Theorem in HOL Light

|- !g h f f’.

simple_path g /\ simple_path h /\

homeomorphism (path_image g,path_image h) (f,f’)

==> ?k k’.

homeomorphism ((:real^2),(:real^2)) (k,k’) /\

(!x. x IN path_image g ==> k x = f x) /\

(!y. y IN path_image h ==> k’ y = f’ y) /\

IMAGE k (path_image g) = path_image h /\

IMAGE k (inside(path_image g)) =

inside(path_image h) /\

IMAGE k (outside(path_image g)) =

outside(path_image h)

Proof uses not just topology but also complex analysis (Riemann
Mapping Theorem, . . . )



Why isn’t the Jordan Curve Theorem obvious?

I Continuous curves can have counterintuitive properties, even
filling a “solid” shape (Peano’s space-filling curves,
Hahn-Mazurkiewicz theorem)

I While simple curves can’t fill 2-D regions, they can still have
2-D Lebesgue density 1− ε (Osgood, Knopp).

I Higher-dimensional analogs in RN look equally plausible. But
while Jordan does extend, Jordan-Schoenflies does not
(Alexander).

Alexander’s Horned Sphere



HOL Light Multivariate library

Partly as a result of Flyspeck, HOL Light is particularly strong in
the area of topology, analysis and geometry in Euclidean space Rn.

File Lines Contents

misc.ml 2540 Background stuff
metric.ml 23331 Metric spaces and general topology
homology.ml 10539 Singular homology
vectors.ml 10799 Basic vectors, linear algebra
determinants.ml 4776 Determinant and trace
topology.ml 36850 Topology of euclidean space
convex.ml 17848 Convex sets and functions
paths.ml 28185 Paths, simple connectedness etc.
polytope.ml 8940 Faces, polytopes, polyhedra etc.
degree.ml 9720 Degree theory, retracts etc.
derivatives.ml 5758 Derivatives
clifford.ml 979 Geometric (Clifford) algebra
integration.ml 26145 Integration
measure.ml 29998 Lebesgue measure



Multivariate theories continued

From this foundation complex analysis is developed and used to
derive convenient theorems for R as well as more topological
results.

File Lines Contents

complexes.ml 2249 Complex numbers
canal.ml 4019 Complex analysis
transcendentals.ml 7559 Real & complex transcendentals
realanalysis.ml 17556 Some analytical stuff on R
moretop.ml 9583 Further topological results
cauchy.ml 23774 Complex line integrals

Credits: JRH, Marco Maggesi, Valentina Bruno, Graziano Gentili,
Gianni Ciolli, Lars Schewe, . . .



The awkward squad

There are some classic theorems saying something quite concrete
about Euclidean space that seem difficult to prove ‘with bare
hands’, though I tried . . .

, Brouwer’s fixed-point theorem: Kuhn’s explicit combinatorial
proof with cubical subdivision

, The hairy ball theorem: ad-hoc partial notion of topological
degree following Dugundji

/ The Jordan-Brouwer separation theorem: homotopic
characterization following Borsuk a partial solution

/ The Borsuk-Ulam theorem: ???

The Jordan-Brouwer Separation Theorem in HOL Light

Thanks to the singular homology theory, we can now derive this
general form of the Jordan Curve Theorem:

|- !s. 2 <= dimindex(:N) /\

s homeomorphic sphere(vec 0,&1)

==> ?ins out.

~(ins = {}) /\ open ins /\ connected ins /\

~(out = {}) /\ open out /\ connected out /\

bounded ins /\ ~bounded out /\

ins INTER out = {} /\

ins UNION out = (:real^N) DIFF s /\

frontier ins = s /\

frontier out = s



Basic idea of homology theory

For each p ∈ Z, define a ‘pth homology group’ functor mapping
topological spaces to groups

I Topological space X is mapped to its homology group Hp(X )

I A continuous map f : X → Y between topological spaces
gives rise to an induced group homomorphism
f∗ : Hp(X )→ Hp(Y )

Functoriality means IX ∗ = IHp(X ) and (f ◦ g)∗ = f∗ ◦ g∗. A
homology theory satisfies other nice properties too, e.g.

I If two maps f : X → Y and g : X → Y are homotopic then
f∗ = g∗

The challenge is to set up such a functor . . .



Top-level steps in the Jordan-Brouwer proof

I ISOMORPHIC_HOMOLOGY_GROUPS_EUCLIDEAN_COMPLEMENTS:
If two closed S ,T ⊆ RN are homeomorphic, their
complements have isomorphic homology groups:
Hp(Rn − S) ∼= Hp(Rn − T ) for all p ∈ Z

I ZEROTH_HOMOLOGY_GROUP: For any space X , the zeroth
homology group H0(X ) is isomorphic to the free abelian group
on the set of path components of X . For singular homology
with integer coefficients, not necessarily all homology theories

I ISOMORPHIC_FREE_ABELIAN_GROUPS: Two free abelian
groups are isomorphic iff their generating sets have the same
cardinality

I Now if S is homeomorphic to the sphere Sn−1, since Sn−1 is
compact so is S and therefore they are both closed so the
theorem applies. Their complements are open and Rn is
locally path-connected, so components and path components
are the same.



Basic source for the first part



Setting up group theory

We first set up a certain amount of basic group theory (group,
homomorphism, isomorphism, quotient group, . . . ) including the
notion of an exact sequence of groups and homomorphisms

. . . Gn Gn+1 . . .
fn−1 fn fn+1

meaning that the image of each map is the same as the kernel of
the following map, im(fn) = ker(fn+1), where for f : G → H these
are defined by

I im(f ) = {f (x) | x ∈ G}
I ker(f ) = {x ∈ G | f (x) = 1H}



Group theory in HOL Light

Image and kernel (with the source and target groups as extra
parameters)

|- group_image (G,G’) f = IMAGE f (group_carrier G)

|- group_kernel (G,G’) f =

{x | x IN group_carrier G /\ f x = group_id G’}

Exactness of a pair of homomorphisms at their ‘interface’ group:

|- group_exactness (G,H,K) (f,g) <=>

group_homomorphism (G,H) f /\

group_homomorphism (H,K) g /\

group_image (G,H) f = group_kernel (H,K) g



Commutative diagrams and diagram chasing
There are common patterns involving exact sequences: the five
lemma, four lemma, snake lemma, or (Dieck’s Algebraic Topology):

These are generally easy to prove even formally by “diagram
chasing” but look ugly and baffling in a purely textual
representation.



Free Abelian groups

A basic ingredient in setting singular homology is the notion of a
free abelian group over a set S , which is intuitively the set of
formal sums

a1x1 + · · ·+ anxn

where ai ∈ Z, and xi ∈ S , with the expected rules for addition,
subtraction and multiplication by integers

(3x + 4y)− 2(2y − x) = 5x

Formally we can regard these as the functions S → Z with finite
support (i.e. giving 0 for all but finitely many elements of S), with
addition etc. pointwise.



Free Abelian groups in HOL Light

In HOL Light we set up a type ‘:A frag’ of free Abelian groups
over a type :A, with associated arithmetic operations frag add,
frag cmul etc., and even a handy decision procedure for basic
algebraic rearrangements:

FRAG_MODULE

‘frag_cmul(a + b) c =

frag_add (frag_cmul a c) (frag_cmul b c)‘;;

Linked to the general formalization of group theory like this

|- free_abelian_group s =

group ({c | frag_support c SUBSET s},

frag_0,

frag_neg,

frag_add)



Standard simplices

For each p ∈ N we define the standard p-simplex to be the subset
of Rp+1 with Σp

0xi = 1, or in HOL Light:

|- standard_simplex p =

{ x:num->real | (!i. &0 <= x i /\ x i <= &1) /\

(!i. p < i ==> x i = &0) /\

sum (0..p) x = &1}

Note that all these live inside the infinite product RN, not any
bounded Euclidean space.



Singular simplices
A singular p-simplex in a topological space X is simply a
continuous function from the standard p-simplex to X :

Singular = function need not be injective so image of a singular
p-simplex can have topological dimension < p. In HOL Light:

|- singular_simplex (p,top) f <=>

continuous_map(subtopology (product_topology (:num)

(\i. euclideanreal))

(standard_simplex p),

top) f /\

EXTENSIONAL (standard_simplex p) f



Singular face
The singular face map takes a singular p-simplex to a singular
(p − 1)-simplex by composing through the k ’th face of the
standard simplex

|- face_map k x =

\i. if i < k then x i

else if i = k then &0 else x(i - 1)

|- singular_face p k f =

RESTRICTION (standard_simplex (p - 1))

(f o face_map k)



Singular chains

Now the singular chain group Cp(X ) for a topological space X is
just the free Abelian group on the singular p-simplices over X :

|- singular_chain (p,top) c <=>

frag_support c SUBSET singular_simplex(p,top)

We will often appeal to the fact that functions over the generating
set of a free Abelian group automatically extend to the whole
group, encoded:

|- frag_extend f x =

iterate frag_add (:A)

(\a. frag_cmul (dest_frag x a) (f a)))



The boundary of a singular chain

We define the boundary of a p-simplex σ to be a sum with
alternating signs of the various faces σk :

Σp
k=0(−1)kσk

This then gives by extension the boundary homomorphism
∂p : Cp(X )→ Cp−1(X ):

|- chain_boundary p c =

if p = 0 then frag_0 else

frag_extend

(\f. iterate frag_add (0..p)

(\k. frag_cmul (--(&1) pow k)

(frag_of(singular_face p k f)))) c



Cycles and boundaries
We define the singular cycles Zp(X ) as the kernel of the boundary
map ∂p and the singular boundaries Bp(X ) as the image of ∂p+1.
More generally we define relative forms Zp(X ,A) and Bp(X ,A)
where we ignore chains in the subtopology for a subset A:

|- mod_subset (p,top) c1 c2 <=>

singular_chain (p,top) (frag_sub c1 c2)

|- singular_relcycle(p,top,s) c <=>

singular_chain (p,top) c /\

(chain_boundary p c == frag_0)

(mod_subset(p-1,subtopology top s))

|- singular_ relboundary(p,top,s) c <=>

?d. singular_chain (p + 1,top) d /\

(chain_boundary (p + 1) d == c)

(mod_subset (p,subtopology top s))



The boundary of a boundary

A fundamental fact (which you can prove by brute force) is that
the boundary of a boundary is zero, ∂p−1 ◦ ∂p = 0:

|- singular_chain (p,top) c

==> chain_boundary (p - 1) (chain_boundary p c) =

frag_0

This implies every (relative) boundary is a (relative) cycle,
Bp(X ,A) ⊆ Zp(X ,A):

|- singular_relboundary(p,top,s) c

==> singular_relcycle(p,top,s) c



The (relative) homology relation

At last we define the notion of relative homology: σ, σ′ ∈ Cp(X )
are homologous relative to A if their difference is a relative
boundary, σ − σ′ ∈ Bp(X ,A):

|- homologous_rel (p,top,s) c1 c2 <=>

singular_relboundary (p,top,s) (frag_sub c1 c2)

Intuitively the difference “is the boundary of a region”.



Induced map

A continuous map f : X → Y between topological spaces gives rise
to an induced group homomorphism f] : Cp(X )→ Cp(Y )

|- simplex_map p g c =

RESTRICTION (standard_simplex p) (g o c)

|- chain_map p g c =

frag_extend (frag_of o simplex_map p g) c



Homology groups at last!
Define actual groups corresponding to Cp(X ) and Zp(X ,A):

|- chain_group (p,top) =

free_abelian_group (singular_simplex (p,top))

|- relcycle_group (p,top,s) =

subgroup_generated (chain_group (p,top))

(singular_relcycle (p,top,s))

Relative homology group Hp(X ,A) is the quotient group
Zp(X ,A)/Bp(X ,A), and Hp(X ) =def Hp(X , ∅).

|- relative_homology_group (p,top,s) =

if p < &0 then singleton_group ARB else

quotient_group (relcycle_group (num_of_int p,top,s))

(singular_relboundary (num_of_int p,top,s)))

|- homology_group(p,top) =

relative_homology_group(p,top,{})



Induced map and boundary map

The induced f] : Cp(X )→ Cp(Y ) and boundary
∂p : Cp(X )→ Cp−1(X ) homomorphisms appropriately respect the
homology relation and can be lifted to the relative homology
groups. Informally we write

I ∂ : Hp(X ,A)→ Hp−1(A)

I f∗ : Hp(X ,A)→ Hp(Y ,B) when f : X → Y is a continuous
map with f [A] ⊆ B

In the actual formalization they are much more heavily
parametrized with p and the topological pairs.



Functoriality

The induced map is a functor: identity map inducing identity
homomorphism and (g ◦ f )∗ = g∗ ◦ f∗, or in HOL Light:

|- continuous_map (top,top’) f /\

IMAGE f s SUBSET t /\

continuous_map (top’,top’’) g /\

IMAGE g t SUBSET u

==> hom_induced p (top,s) (top’’,u) (g o f) =

hom_induced p (top’,t) (top’’,u) g o

hom_induced p (top,s) (top’,t) f)

The heavy parametrization and conditions mean there’s often a lot
of detail to fill in relative to textbook presentations.



Naturality

The boundary map is a natural transformation, i.e. ∂ ◦ f∗ = f∗ ◦ ∂

|- continuous_map (top,top’) f /\ IMAGE f s SUBSET t

==> hom_boundary p (top’,t) o

hom_induced p (top,s) (top’,t) f =

hom_induced (p - &1) (subtopology top s,{})

(subtopology top’ t,{}) f o

hom_boundary p (top,s))

Again a much more verbose statement informally buried in the
assumption that f is a map of topological pairs.



The Eilenberg-Steenrod Axioms

Most applications of homology theory just need a number of basic
properties codified by Eilenberg, Steenrod and Milnor. As well as
functoriality and naturality:

I If f and g are homotopic then f∗ = g∗.

I Excision: if closure(U) ⊆ interior(A) then the map
ι∗ : Hp(X − U,A− U)→ Hp(X ,A) induced by inclusion is an
isomorphism.

I Dimension: if X is a 1-point space then Hp(X ) is the trivial
group for p 6= 0.

I Exactness of homology sequence

Hp(A) Hp(X ) Hp(X ,A) Hp−1(A)∂ i∗ j∗ ∂

I Additivity: added later by Milnor, follows from others for finite
sums.



Two of the Eilenberg-Steenrod axioms in HOL Light
The two most difficult of the axioms to prove are homotopy (via a
prism construction)

|- homotopic_with (\h. IMAGE h s SUBSET t) (top,top’) f g

==> hom_induced p (top,s) (top’,t) f =

hom_induced p (top,s) (top’,t) g

and excision (via repeated subdivision of a complex)

|- top closure_of u SUBSET top interior_of t /\

t SUBSET s

==> group_isomorphism

(relative_homology_group

(p,subtopology top (s DIFF u),t DIFF u),

relative_homology_group

(p,subtopology top s,t))

(hom_induced p

(subtopology top (s DIFF u),t DIFF u)

(subtopology top s,t) (\x. x))



Applications of homology

HOL Light development is about 5700 lines to get the E-S-M
axioms and another 5000 developing consequences of the axioms
and making applications:

I The generalized Jordan curve theorem, following Dold (see
above)

I Easy development of topological degree from the fact that
spheres have (reduced) homology groups isomorphic to Z so
the induced map is just multiplication by an integer, hence
other consequences like Brouwer.

I A proof of the Borsuk-Ulam theorem following Dieck’s
Algebraic Topology with a mix of homology and a little bit of
homotopy to deform a sphere map.

Lots more interesting stuff to do, e.g. cohomology, homology with
other coefficients and the Lefschetz fixed-point theorem, Alexander
duality, . . .



Thank you!


